VA for NLS soliton using a Gaussian ansatz inside a parabolic potential

```
> restart;
> interface(showassumed=0):;
```

Ansatz of the form u(x,t) = A(x,t) * exp(I*phi(x,t))

```
> A := B(t)*exp(-(x-xi(t))^2/(2*w(t)^2));
> phi := d(t)*(x-xi(t))^2+c(t)*(x-xi(t))+b(t);
```

$$A := B(t)\, \mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}$$

$$\phi := d(t)\, (x-\xi(t))^2 + c(t)\, (x-\xi(t)) + b(t) \qquad \textbf{(1)}$$

Potential

```
> V := Omega^2/2*x^2;
```

$$V := \frac{\Omega^2\, x^2}{2} \qquad \textbf{(2)}$$

Lagrangian: (I/2)(u u*_t - u_t u*) + (1/2) u_x u*_x - (1/2) |u|^4 = A^2 phi_t + (1/2)*(A_x^2+A^2 phi_x^2) -(1/2)*A^4+V*A^2

```
> LA := A^2*diff(phi,t) + (1/2)*(diff(A,x)^2+A^2*diff(phi,x)^2) -
  (1/2)*A^4 + V*A^2;
```

$$LA := B(t)^2 \left(\mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}\right)^2 \left(\left(\frac{\mathrm{d}}{\mathrm{d}t}\, d(t)\right)(x-\xi(t))^2 - 2\, d(t)\, (x-\xi(t)) \left(\frac{\mathrm{d}}{\mathrm{d}t}\, \xi(t)\right) + \left(\frac{\mathrm{d}}{\mathrm{d}t}\, c(t)\right)(x-\xi(t)) - c(t) \left(\frac{\mathrm{d}}{\mathrm{d}t}\, \xi(t)\right) + \frac{\mathrm{d}}{\mathrm{d}t}\, b(t)\right) + \frac{B(t)^2\, (x-\xi(t))^2 \left(\mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}\right)^2}{2\, w(t)^4} + \frac{B(t)^2 \left(\mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}\right)^2 (2\, d(t)\, (x-\xi(t)) + c(t))^2}{2} - \frac{B(t)^4 \left(\mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}\right)^4}{2} + \frac{\Omega^2\, x^2\, B(t)^2 \left(\mathrm{e}^{-\frac{(x-\xi(t))^2}{2\, w(t)^2}}\right)^2}{2} \qquad \textbf{(3)}$$

Take out t-dependence so we can compute Leff

```
> sub1 := diff(B(t),t)=Bp,diff(d(t),t)=dp,diff(c(t),t)=cp,diff(w
  (t),t)=wp,diff(b(t),t)=bp,diff(xi(t),t)=xip:
> sub2 := xi(t)=xi,b(t)=b,B(t)=B,c(t)=c,d(t)=d,w(t)=w:
> LA2 := subs({sub1,sub2},LA);
```

$$LA2 := B^2 \left(e^{-\frac{(x-\xi)^2}{2w^2}}\right)^2 \left(dp\,(x-\xi)^2 - 2\,d\,(x-\xi)\,xip + cp\,(x-\xi) - c\,xip + bp\right) + \frac{B^2 (x-\xi)^2 \left(e^{-\frac{(x-\xi)^2}{2w^2}}\right)^2}{2\,w^4} + \frac{B^2 \left(e^{-\frac{(x-\xi)^2}{2w^2}}\right)^2 (2\,d\,(x-\xi)+c)^2}{2} - \frac{B^4 \left(e^{-\frac{(x-\xi)^2}{2w^2}}\right)^4}{2} + \frac{\Omega^2 x^2 B^2 \left(e^{-\frac{(x-\xi)^2}{2w^2}}\right)^2}{2} \quad \textbf{(4)}$$

Leff = integral of Lag, we need to assume that a>0 and xi real to be able to evaluate integrals

```
> assume(w>0);
> assume(c>0);
> assume(xi>0);
> assume(b>0);
> assume(d>0);
> Leff := int(LA2,x=-infinity..infinity);
```

$$Leff := B^2 (-c\,xip + bp)\,w\sqrt{\pi} + \frac{B^2\,dp\,w^3\sqrt{\pi}}{2} + \frac{B^2\sqrt{\pi}}{4\,w} + \frac{B^2 c^2 w\sqrt{\pi}}{2} + B^2 d^2 w^3\sqrt{\pi} - \frac{B^4\sqrt{2}\,w\sqrt{\pi}}{4} + \frac{\Omega^2 \xi^2 B^2 w\sqrt{\pi}}{2} + \frac{\Omega^2 B^2 w^3\sqrt{\pi}}{4} \quad \textbf{(5)}$$

Put back t-dependences so that we can do Euler-Lag

```
> subi1 := Bp=diff(B(t),t),dp=diff(d(t),t),cp=diff(c(t),t),
  wp=(diff(w(t),t)),bp=(diff(b(t),t)),xip=(diff(xi(t),t)):
> subi2 := xi=xi(t),b=b(t),B=B(t),d=d(t),c=c(t),w=w(t):
```

Euler-Lagrange eqs

```
> dLwp := subs({subi1,subi2},diff(Leff,wp)):;
> dLbp := subs({subi1,subi2},diff(Leff,bp)):;
> dLcp := subs({subi1,subi2},diff(Leff,cp)):;
> dLdp := subs({subi1,subi2},diff(Leff,dp)):;
> dLBp := subs({subi1,subi2},diff(Leff,Bp)):;
> dLxip:= subs({subi1,subi2},diff(Leff,xip)):;
> eq1 := diff(dLwp ,t)=subs({subi1,subi2},diff(Leff,w));
> eq2 := diff(dLbp ,t)=subs({subi1,subi2},diff(Leff,b));
> eq3 := diff(dLcp ,t)=subs({subi1,subi2},diff(Leff,c));
> eq4 := diff(dLdp ,t)=subs({subi1,subi2},diff(Leff,d));
> eq5 := diff(dLBp ,t)=subs({subi1,subi2},diff(Leff,B));
> eq6 := diff(dLxip,t)=subs({subi1,subi2},diff(Leff,xi));
```

$$eq1 := 0 = B(t)^2 \left(-c(t)\left(\frac{d}{dt}\xi(t)\right) + \frac{d}{dt}b(t)\right)\sqrt{\pi} + \frac{3\,B(t)^2 \left(\frac{d}{dt}d(t)\right) w(t)^2\sqrt{\pi}}{2}$$

$$-\frac{B(t)^2\sqrt{\pi}}{4\,w(t)^2}+\frac{B(t)^2\,c(t)^2\sqrt{\pi}}{2}+3\,B(t)^2\,d(t)^2\,w(t)^2\sqrt{\pi}-\frac{B(t)^4\sqrt{2}\sqrt{\pi}}{4}$$

$$+\frac{\Omega^2\,\xi(t)^2\,B(t)^2\sqrt{\pi}}{2}+\frac{3\,\Omega^2\,B(t)^2\,w(t)^2\sqrt{\pi}}{4}$$

$$eq2 := 2\,B(t)\,w(t)\sqrt{\pi}\left(\frac{d}{dt}\,B(t)\right)+B(t)^2\left(\frac{d}{dt}\,w(t)\right)\sqrt{\pi}=0$$

$$eq3 := 0=-B(t)^2\left(\frac{d}{dt}\,\xi(t)\right)w(t)\sqrt{\pi}+B(t)^2\,c(t)\,w(t)\sqrt{\pi}$$

$$eq4 := B(t)\,w(t)^3\sqrt{\pi}\left(\frac{d}{dt}\,B(t)\right)+\frac{3\,B(t)^2\,w(t)^2\sqrt{\pi}\left(\frac{d}{dt}\,w(t)\right)}{2}$$

$$=2\,B(t)^2\,d(t)\,w(t)^3\sqrt{\pi}$$

$$eq5 := 0=2\,B(t)\left(-c(t)\left(\frac{d}{dt}\,\xi(t)\right)+\frac{d}{dt}\,b(t)\right)w(t)\sqrt{\pi}+B(t)\left(\frac{d}{dt}\right.$$

$$\left.d(t)\right)w(t)^3\sqrt{\pi}+\frac{B(t)\sqrt{\pi}}{2\,w(t)}+B(t)\,c(t)^2\,w(t)\sqrt{\pi}+2\,B(t)\,d(t)^2\,w(t)^3\sqrt{\pi}$$

$$-B(t)^3\sqrt{2}\,w(t)\sqrt{\pi}+\Omega^2\,\xi(t)^2\,B(t)\,w(t)\sqrt{\pi}+\frac{\Omega^2\,B(t)\,w(t)^3\sqrt{\pi}}{2}$$

$$eq6 := -2\,B(t)\,c(t)\,w(t)\sqrt{\pi}\left(\frac{d}{dt}\,B(t)\right)-B(t)^2\left(\frac{d}{dt}\,c(t)\right)w(t)\sqrt{\pi} \tag{6}$$

$$-B(t)^2\,c(t)\left(\frac{d}{dt}\,w(t)\right)\sqrt{\pi}=\Omega^2\,\xi(t)\,B(t)^2\,w(t)\sqrt{\pi}$$

Sove Euler-Lag ODEs simultaneously

```
> sol:=solve({eq1,eq2,eq3,eq4,eq5,eq6},{diff(B(t),t),diff(d(t),t),
  diff(c(t),t),diff(w(t),t),diff(b(t),t),diff(xi(t),t)}):
> eqw := diff(w(t),t)=subs(sol,diff(w(t),t));
> eqb := diff(b(t),t)=subs(sol,diff(b(t),t));
> eqc := diff(c(t),t)=subs(sol,diff(c(t),t));
> eqd := diff(d(t),t)=subs(sol,diff(d(t),t));
> eqB := diff(B(t),t)=subs(sol,diff(B(t),t));
> eqxi := diff(xi(t),t)=subs(sol,diff(xi(t),t));
```

$$eqw := \frac{d}{dt}\,w(t)=2\,w(t)\,d(t)$$

$$eqb := \frac{d}{dt}\,b(t)=-\frac{4\,\Omega^2\,\xi(t)^2\,w(t)^2-5\,B(t)^2\sqrt{2}\,w(t)^2-4\,c(t)^2\,w(t)^2+4}{8\,w(t)^2}$$

$$eqc := \frac{d}{dt}\,c(t)=-\Omega^2\,\xi(t)$$

$$eqd := \frac{d}{dt}\, d(t) = -\frac{8\, d(t)^2\, w(t)^4 + 2\, \Omega^2\, w(t)^4 + B(t)^2 \sqrt{2}\, w(t)^2 - 2}{4\, w(t)^4}$$

$$eqB := \frac{d}{dt}\, B(t) = -B(t)\, d(t)$$

$$eqxi := \frac{d}{dt}\, \xi(t) = c(t) \qquad (7)$$

Obtain equation for position

```
> ode_xi := subs({eqc},diff(eqxi,t));
```

$$ode\_xi := \frac{d^2}{dt^2}\, \xi(t) = -\Omega^2\, \xi(t) \qquad (8)$$

Obtain quation for width

```
> ode_w := diff(w(t),t,t)=simplify(subs({eqd,eqw},diff(rhs(eqw),t))
  );
```

$$ode\_w := \frac{d^2}{dt^2}\, w(t) = \frac{-2\, \Omega^2\, w(t)^4 - B(t)^2 \sqrt{2}\, w(t)^2 + 2}{2\, w(t)^3} \qquad (9)$$

Now apply conservation of mass to solve for B(t) as a function of w(t);

```
> A2 := subs({sub1,sub2},A^2):;
> Mass := int(A2,x=-infinity..infinity);
> BM := solve(Mass=M,B):;
> BMp := subs({w=w(t)},BM[1]);
```

$$Mass := B^2\, w \sqrt{\pi}$$

$$BMp := \frac{\sqrt{w(t) \sqrt{\pi}\, M}}{w(t) \sqrt{\pi}} \qquad (10)$$

Width equation with conservation of mass

```
> ode_w2:=expand(simplify(subs(B(t)=BMp,ode_w)));
```

$$ode\_w2 := \frac{d^2}{dt^2}\, w(t) = -w(t)\, \Omega^2 - \frac{M \sqrt{2}}{2 \sqrt{\pi}\, w(t)^2} + \frac{1}{w(t)^3} \qquad (11)$$